





# Omron automata programagarriak

## PRAKTIKAK



LANBIDE  
EKIMENA



▣ *Proiektuaren bultzatzaileak*



▣ *Laguntzaileak*



Gipuzkoako Foru Aldundia  
Diputación Foral de Gipuzkoa  
Gizarte eta Erakunde Harremanetarako  
Departamentua

▣ *Hizkuntz koordinazioa*



hizkuntz  
ELHUYAR  
zerbitzuak

*Egilea(k)*: LORES ESPINOSA, JOSE CARLOS

*Zuzenketak*: Elhuyar Hizkuntza Zerbitzuak

*Maketa*: Ainara Sarasketa

*Azalaren diseinua*: Naiara Beasain

**2006an prestatua**



## Aurkibidea

1.	PRAKTIKA. LD, OUT, AND, OR ETA NOT INSTRUKZIOAK.....	2
1.1.	LD, OUT.....	3
1.2.	AND, OR.....	4
1.3.	NOT .....	6
1.4.	Ariketak.....	6
2.	PRAKTIKA. AND LD ETA OR LD INSTRUKZIOAK .....	16
2.1.	AND LD.....	16
2.2.	OR LD.....	18
3.	PRAKTIKA. TENPORIZADOREAK, TIM INSTRUKZIOA.....	20
3.1.	Ariketak.....	22
4.	PRAKTIKA. KONTAGAILUAK, CNT INSTRUKZIOA .....	33
4.1.	Ariketak.....	35
5.	PRAKTIKA .....	40
5.1.	Barne-erreleak .....	40
5.2.	Barne-errele memoriadunak.....	42
6.	PRAKTIKA. OUT-OUT NOT, SET-RESET, KEEP (11), DIFU (13), DIFD (14).....	43
6.1.	OUT, OUT NOT.....	43
6.2.	SET, RESET .....	43
6.3.	KEEP (11).....	44
6.4.	DIFU (13), DIFD (14).....	46
7.	PRAKTIKA. INTERLOCK IL(02), INTERLOCK CLEAR ILC(03) .....	49
8.	PRAKTIKA. JMP(04)-JME(05).....	52
9.	PRAKTIKA. DIAGNOSTIC FUNTZIOA. FAL(06), FALS(07) .....	54
10.	ARIKETAK .....	60

---

# **OMRON AUTOMATA PROGRAMAGARRIAK PRAKTIKAK**

---

## 1 PRAKTIKA. LD, OUT, AND, OR ETA NOT INSTRUKZIOAK

Praktikak egiteko automata programagarri hau erabili dugu: OMRON CPM2A. Automata programatzeko, memoriaren helbideak hartu behar dira kontuan. Sarreran, kaptadoreak egoten dira (sakagailuak, ibiltarteko etengailuak, sentsoreak, fotozelulak...). Horiek konektatzeko edo simulatzeko, hamar helbide daude IR memoriaren barruan, IR000tik IR009ra hain zuzen ere. Irteeran, berriz, eragingailuak egoten dira (motorrak, lanparak, erreleak...): eta horiek konektatzeko edo simulatzeko, beste hamar helbide daude IR memoriaren barruan, IR 010etik IR019ra hain zuzen.

Sarrerak honela programatu behar dira:

0. sarrera: 000.00
1. sarrera: 000.01
2. sarrera: 000.02
3. sarrera: 000.03

Irteerak, ordea, beste era honetan programatu behar dira:

0. irteera: 10.00
1. irteera: 10.01
2. irteera: 10.02
3. irteera: 10.03

Programa guztien amaieran END(01) funtzioa jarri behar da.

## 1.1 LD, OUT

LD instrukzioak beste adar bat irekitzen du zirkuituan.

OUT instrukzioak irteera aktibatzen du: hots, zirkuituaren amaiera da.

### 1. adibidea (LD, OUT)

Etengailu baten bidez lanpara bat kontrolatzea da helburua.

**Sarrera:** 000.00, etengailua.

**Irteera:** 10.00, lanpara.

Etengailua ixtean, piztu egiten da lanpara.


**Programa:**

LD 000.00

OUT 10.00

END(01)

**Eskema:**



## 1.2 AND, OR

AND instrukzioak seriean jartzen ditu bi kontaktu.

OR instrukzioak, berriz, paraleloan jartzen ditu bi kontaktu.

### 2. adibidea (AND)

Bi etengailuren bidez lanpara bat kontrolatzea da helburua.

Bi etengailuak itxita daudenean, piztu egiten da lanpara; bietako bat irekita baldin badago, lanpara itzalita dago.


**Sarrerak:** 000.00, 0 etengailua.  
000.01, 1 etengailua.

**Irteera:** 10.00, lanpara.

#### Programa:

```
LD 000.00
AND 000.01
OUT 10.00
END(01)
```

#### Eskema:





### 3. adibidea (OR)

Bi etengailuren bidez lanpara bat kontrolatzea da helburua.

Bi etengailuetako bat itxita baldin badago, piztu egiten da lanpara; biak irekita baldin badaude, lanpara itzalita dago.

**Sarrerak:** 000.00, 0 etengailua.

000.01, 1 etengailua.

**Irteera:** 10.00, lanpara.

**Programa:**


LD 000.00

OR 000.01

OUT 10.00

END(01)

**Eskema:**



### 1.3 NOT

NOT instrukzioak kontaktua aldatzen du (itxita/irekita).

### 1.4 Ariketak

#### 1. ariketa

Etengailu baten bitartez motor bat kontrolatzea da helburua. Etengailua ixtean, martxan jartzen da motorra. Motorra funtzionatzen ari dela seinaleztatzeko bi lanpara erabiliko ditugu. Etengailua ixten dugunean, martxan jartzen da motorra eta, aldi berean, bi lanparok piztu egiten dira.


**Sarrera:** 000.00, etengailua (martxan jartzekoa).

**Irteerak:** 10.00, motorra.  
10.01, lanpara gorria.  
10.02, lanpara urdina.

**Programa:**

```
LD 000.00
OUT 10.00
OUT 10.01
OUT 10.02
END(01)
```

**Eskema:**



2. ariketa

**Sarrera:** 000.00, etengailua.

000.01, sentsorea.

**Irteera:** 10.00, A motorra.


10.01, B motorra.

Etengailua ixten dugunean, A motorra martxan jartzen da. Gainera, sentsorea aktibatzen bada, B motorra ere martxan jartzen da.

**Programa:**

```
LD 000.00
OUT 10.00
AND 000.01
OUT 10.01
END(01)
```

**Eskema:**



### 3. ariketa

**Sarrerak:** 000.00, 0 etengailua.  
 000.01, 1 etengailua.  
 000.02, 2 etengailua.  
 000.03, 3 etengailua.  
 000.04, 4 etengailua.


**Irteerak:** 10.00, lanpara.  
 10.01, txirrina.  
 10.02, motorra.

0 etengailua ixten dugunean, lanpara piztu egiten da; gainera, 1 etengailua ere ixten badugu, txirrina jotzen hasten da. Bestalde, beste hiru etengailuak ere ixten baditugu, motorra martxan jartzen da.

**Programa:**

```
LD 000.00
OUT 10.00
AND 000.01
OUT 10.01
AND 000.02
AND 000.03
AND 000.04
OUT 10.02
END(01)
```

**Eskema:**



#### 4. ariketa

**Sarrerak:** 000.00, A sentsorea.

000.01, B sentsorea.

000.02, etengailua.

**Irteerak:** 10.00, A sentsorearen seinaleztapena (A lanpara).

10.01, B sentsorearen seinaleztapena (B lanpara).

10.02, etengailuaren seinaleztapena (E lanpara).

10.03, botilak betetzekoa.

10.04, botilak betetzekoaren seinaleztapena (D lanpara).

#### Funtzionamendua:

Botilak likidoz betetzeko prozesua da. Sentsoreek botila behar bezala jarrita dagoela adierazten dute. Orduan, langileak etengailua itxi eta botila bete egiten da. Sentsore bietako bat ez badago aktibatuta, botilak betetzeko makina ez da martxan jartzen, nahiz eta langileak etengailua itxi. Hau da, botilak likidoz betetzeko makina martxan jartzeko, sentsore biak eta etengailua aktibatu behar dira. Horietakoren bat ez badago aktibatuta, makina ez da martxan jartzen.

A sentsorea aktibatzen denean, A lanpara pizten da seinaleztatzeko.

B sentsorea aktibatzen denean, B lanpara pizten da seinaleztatzeko.

Etengailua itxita dagoenean, E lanpara pizten da seinaleztatzeko.

Botilak betetzeko makina martxan dagoenean, D lanpara pizten da seinaleztatzeko.

#### Programa:

LD 000.00

OUT 10.00

LD 000.01

OUT 10.01

LD 000.02

OUT 10.02

LD 10.00

AND 10.01


AND 10.02

OUT 10.03

OUT 10.04

END (01)

Eskema:



## 5. ariketa


**Sarrerak:** 000.00, A sentsorea.  
 000.01, B sentsorea.  
 000.02, etengailua.

**Irteera:** 10.00, motorra.

A sentsorea aktibatuta eta etengailua itxita baldin badaude, motorra martxan jartzen da.  
 B sentsorea aktibatuta eta etengailua itxita baldin badaude, motorra martxan jartzen da.

**Programa:**

```
LD 000.00
OR 000.01
AND 000.02
OUT 10.00
END(01)
```

**Eskema:**

## 6. ariketa

**Sarrerak:** 000.00, 0 etengailua.  
 000.01, A sentsorea.  
 000.02, 2 etengailua.  
 000.03, B sentsorea.

**Irteerak:** 10.00, errelea.


Errelea honako hauetan aktibatzen da:

- 0 etengailua, A sentsorea eta 2 etengailua aktibatuta daudenean.
- B sentsorea eta 2 etengailua aktibatuta daudenean.

### Programa:

```
LD 000.00
AND 000.01
OR 000.03
AND 000.02
OUT 10.00
END(01)
```

### Eskema:





## 7. ariketa

Motor bat kontrolatzea da helburua. Errele termikoa eta stop-sakagailua posizio egokian badaude, martxa-sakagailua sakatzen dugunean, motorra martxan jartzen da. Behin martxan jarrita, errele termikoa edo stop-sakagailua aktibatzen badugu, motorra gelditu egiten da.

**Sarrerak:** 000.00, martxa-sakagailua, normalki irekia.

000.01, stop-sakagailua, normalki itxia.

000.02, errele termikoa, normalki itxia.

**Irteerak:** 10.00, motorra.

**Programa:**

LD 000.00


OR 10.00

AND NOT 000.01

AND NOT 000.02

OUT 10.00

END(01)

**Eskema:**

## 8. ariketa

Motor bat kontrolatzea da helburua. Motorra eskuinaldera nahiz ezkerraldera biratu daiteke. Biraketaren noranzkoa aldatzeko, gelditu egin behar da motorra. Hau da, lehenengo motorra gelditu, eta, gero, biraketaren noranzkoa aukeratu behar da.

Motorra kontrolatzeko, hiru sakagailu erabiltzen dira: eskuinaldera biratzekoa, ezkerraldera biratzekoa eta gelditzekoa (stop-sakagailua).

Motorra babesteko, errele termikoa erabiltzen da.

**Sarrerak:** 000.00, ezkerralderako sakagailua.

000.01, eskuinalderako sakagailua.

000.02, stop-sakagailua.

000.03, errele termikoa.

**Irteerak:** 10.00, ezkerralderako biraketa.

10.01, eskuinalderako biraketa.

### Programa:

LD 000.00

OR 10.00

AND NOT 000.02

AND NOT 000.03

AND NOT 10.01

OUT 10.00

LD 000.01

OR 10.01

AND NOT 000.02

AND NOT 000.03

AND NOT 10.00

OUT 10.01

END(01)

**Eskema:**



Horrelako programa bat erabiltzean, motorra noranzko batean biratzen denean, nahiz eta beste noranzkoan biratzeko agindua jaso, ez dio aginduari jaramonik egiten harik eta geldirik egon arte.

## 2 PRAKTIKA. AND LD ETA OR LD INSTRUKZIOAK

### 2.1 AND LD

Programan sartutako azken bi instrukzio-multzoak seriean batzen ditu AND LD instrukzioak. Zirkuitu batzuek zenbait kontaktu-multzo dituzte seriean konektatuta. Multzoz multzo programatu behar dira zirkuitu horiek, eta AND LD instrukzioak batu egiten ditu programatutako kontaktu-multzoak.

#### 4. adibidea.

**Sarrerak:** 000.00, 0 etengailua  
 000.01, 1 etengailua  
 000.02, 2 etengailua  
 000.03, 3 etengailua

**Irteera:** 10.00, motorra

- 1. multzoa: 0 etengailua eta 3 etengailua, paraleloan konektatuak.
- 2. multzoa: 1 etengailua eta 2 etengailua, paraleloan konektatuak.
- 1. multzoa eta 2. multzoa seriean daude konektatuta.

Motorra egoera hauetan jartzen da martxan:


- 0 etengailua itxita dagoenean eta 1 etengailua edo 2 etengailua itxita daudenean.
- 3 etengailua itxita dagoenean eta 1 etengailua edo 2 etengailua itxita daudenean.

Lehenik, 1. multzoa programatu behar da; gero, bigarrena; eta azkenik, AND LD erabiliz, seriean konektatu behar dira bi multzoak.

**Programa:**

```
LD 000.00
OR 000.03
LD 000.01
OR 000.02
AND LD
OUT 10.00
END(01)
```

Eskema:



## 2.2 OR LD

Programan sartutako azken bi instrukzio-multzoak paraleloan batzen ditu OR LD instrukzioak. Zirkuitu batzuek zenbait kontaktu-multzo dituzte paraleloan konektatuta. Multzoz multzo programatu behar zirkuitu horiek, eta OR LD instrukzioak batu egiten ditu programatutako kontaktu-multzoak.

### 5. adibidea.

**Sarrerak:** 000.00, 0 etengailua  
000.01, 1 etengailua  
000.02, 2 etengailua  
000.03, 3 etengailua

**Irteera:** 10.00, motorra.

1. multzoa: 0 etengailua eta 1 etengailua, seriean konektatuak.
2. multzoa: 2 etengailua eta 3 etengailua, seriean konektatuak.
1. multzoa eta 2. multzoa paraleloan daude konektatuta.

Motorra ondoko hauetan jartzen da martxan:


- 0 etengailua eta 1 etengailua itxita daudenean.
- 2 etengailua eta 3 etengailua itxita daudenean.

Lehenengo multzoa programatu behar da lehendabizi; gero, bigarrena; eta azkenik, OR LD erabiliz, paraleloan konektatu behar dira bi multzoak.

**Programa:**

```
LD 000.00
AND 000.01
LD 000.02
AND 000.03
OR LD
OUT 10.00
END(01)
```

Eskema:



### 3 PRAKTIKA. TENPORIZADOREAK, *TIM* INSTRUKZIOA

TIM instrukzioaren bidez, kontrola daiteke zenbat denbora pasatzen den agindua ematen dugunetik tenporizadoreari dagokion kontaktua itxi arte. Tenporizadoreak duen sarreratik, kontaktzen hasteko agindua bidaltzen da. Tenporizadoreak kontaktzen amaitzen duenean, itxi egin behar da haren kontaktua. Aldi berean, programatuta duen denbora-tartea berriro kontaktzen hasteko prest izango da tenporizadorea (reset).

Tenporizadorea programatzeko, automataren DM memoria erabili behar da, TC 000tik TC 255era, alegia. Kontagailuek ere memoria-eremu hori erabiltzen dutenez gero, programa berean ezin dira zenbaki berak erabiliz izendatu.

Tenporizadoreari izena jarri ostean (TIM 000, TIM 001, ..., TIM 015, ..., TIM255), zehaztu behar da zein denbora-tarte kontatuko duen. Denbora programatzeko, zenbakiaren aurrean “#” jarri eta unitate gisa 0,1 segundo (segundo hamarrena) erabili behar da. Hau da, TIM 000 tenporizadorea hamar segundo kontatzeko programatu nahi bada, datu hauek idatzi behar dira: TIM 000 #0100.

#### 6. adibidea

**Sarrera:** 000.00, martxan jartzeko etengailua.

**Irteera:** 10.00, lanpara.

**Tenporizadorea:** TIM 000, bost segundo.

**Programa:**

LD 000.00

TIM 000

#0050

LD TIM 000

OUT 10.00


END(01)

Martxan jartzeko etengailua ixten denean, tenporizadorea kontaktzen hasten da. Programatutako denbora-tartea (bost segundo) pasatutakoan, tenporizadorearen kontaktua itxi, eta lanpara piztu egiten da.



Martxan jartzeko etengailua irekiz gero, tenporizadoreak kontaktzen amaitu baino lehen, zirkuitua hasierako posiziora itzultzen da, eta, beraz, programatutako denbora-tartea (bost segundo) berriro kontaktzen hasteko prest dago orduan tenporizadorea.

**Eskema:**



### 3.1 Ariketak

#### 9. ariketa

Zer egiten du zirkuitu honek?

**Sarrera:** 000.00, etengailua.


**Irteera:** 10.00, lanpara.

**Tenporizadorea:** TIM 000, bost segundo.

**Programa:**

```
LD 000.00
TIM 000
    #0050
LD NOT TIM 000
OUT 10.00
```

**Eskema:**



Zirkuitua programatu ondoren, PCtik PLCra bidali behar da programa. PLC, RUN posizioan jartzen den unean, piztu egiten da lanpara. 0 etengailua ixtean, tenporizadorea kontaktzen hasten da. Programatutako denbora-tartea (bost segundo) pasatutakoan, TIM 000 kontaktua aktibatu eta lanpara itzali egiten da.

10. ariketa

Zer egiten du zirkuitu honek?

**Sarrera:** 000.00, etengailua.


**Irteera:** 10.00, lanpara.

**Tenporizadorea:** TIM 000, bost segundo.

**Programa:**

```
LD 000.00
TIM 000
    #0050
LD NOT TIM 000
AND 000.00
OUT 10.00
END(01)
```

**Eskema:**



0 etengailua ixten denean, tenporizadorea kontaktzen hasi eta lanpara piztu egiten da. Programatutako denbora-tartea (bost segundo) pasatutakoan, TIM 000 kontaktua aktibatu eta lanpara itzali egiten da.

## 11. ariketa

Zirkuitu hau diseinatu behar da:

Etengailua ixtean, lanparak piztu egin behar du.

Etengailua irekiz gero, lanparak hiru segundo eman behar ditu piztuta, eta gero, itzali egin behar du.

**Sarrera:** 000.00, etengailua.


**Irteera:** 10.00, lanpara.

**Tenporizadorea:** TIM 000, hiru segundo.

### Programa:

```
LD 000.00
OR 10.00
AND NOT TIM 000
OUT 10.00
LD 10.00
AND NOT 000.00
TIM 000
    #0030
END(01)
```

### Eskema:



12. ariketa

Zirkuitu hau diseinatu behar da:

Zirkuitu honen bidez, lanpara bat kontrolatu nahi da. Pizteko agindua ematen zaionetik lau segunora piztu behar du lanparak, hiru segundo eman behar ditu piztuta, eta, azkenik, itzali egin behar du.

**Sarrera:** 000.00, etengailua.

**Irteera:** 10.00, lanpara.


**0 tenporizadorea:** TIM 000, lau segundo.

**1 tenporizadorea:** TIM 001, hiru segundo.

**Programa:**

```
LD 000.00
AND NOT TIM 001
TIM 000
    #0040
LD TIM 000
TIM 001
    #0030
LD TIM 000
OUT 10.00
END(01)
```

**Eskema:**



### 13. ariketa

Zirkuitu hau diseinatu behar da:

Lanpara bi kontrolatu behar dira. Pizteko agindua ematean, A lanparak piztu egin behar du. Handik bi segundora, A lanpara itzali eta B lanparak piztu behar du. Beste bi segundo geroago, B lanparak ere itzali egin behar du.

**Sarrera:** 000.00, etengailua.

**Irteerak:** 10.00, A lanpara.

10.01, B lanpara.

**0 tenporizadorea:** TIM 000, bi segundo.

**1 tenporizadorea:** TIM 001, lau segundo.

**Programa:**

LD 000.00

TIM 000

#0020

TIM 001

#0040

LD NOT TIM 000

AND 000.00

OUT 10.00

LD NOT TIM 001


AND TIM 000

AND 000.00

OUT 10.01

END(01)

Eskema:



## 14. ariketa

Zirkuitu hau diseinatu behar da:

**Sarrerak:** 000.00, martxa-sakagailua.  
000.01, errele termikoa.  
000.02, stop-sakagailua.

**Irteerak:** 10.00, martxan jartzeko seinaleztapena.  
10.01, triangelu-konexioa.  
10.02, izar-konexioa.

**Tenporizadorea:** TIM 000, bost segundo.

### Funtzionamendua:

Martxan jarri behar da motorra. Lehenengo bost segundoetan triangelu-konexioaren bidez, eta gero izar-konexioaren bidez. Triangelu-konexioaren nahiz izar-konexioaren bidez martxan dabilen bitartean, martxa-seinaleztapenak piztuta egon behar du.


Martxa-sakagailua sakatzean, motorrak martxan jarri behar du triangelu-konexioa erabilita: hau da, 10.00 eta 10.01 irteerek aktibatuta egon behar dute. Aldi berean, tenporizadorea kontaktzen hasiko da. Bost segundo pasa ostean, tenporizadorearen kontaktua itxi, triangelu-konexioa desaktibatu eta izar-konexioak aktibatu behar du: hau da, 10.01 konexioa itzali eta 10.02 konexioak piztu behar du. Martxa-seinaleztapenak (10.00) piztuta egon behar du.

### Programa:

```
LD 000.00
OR 10.00
AND NOT 000.01
AND NOT 000.02
OUT 10.00
LD 10.00
TIM 000
    #0050
LD 10.00
AND NOT TIM 000
AND NOT 10.02
OUT 10.01
LD 10.00
AND TIM 000
AND NOT 10.01
OUT 10.02
END(01)
```



Eskema:



## 15. ariketa

Lantegi batean metalezko piezak egiten dituzte. Langileek, pieza bat amaitzen dutenean, garraio-zinta batean uzten dute. Metal-detektagailuak pieza bat detektatzen duen bakoitzean, gelditu egiten da garraio-zinta. Segundo bat geroago, lixagailuak martxan jarri behar du. Lixagailuak behar bezain neurri zehatzetan moldatu behar du pieza. Neurri-detektagailuak piezak neurri zehatzak dituela ziurtatzen dutenean, lixagailuak gelditu eta hasierako posiziora itzuli behar du. Segundo bat geroago, garraio-zintak martxan jarri behar du berriro eta guztiz amaituta dagoen pieza beste garraio-zinta batera eraman behar du.

**Sarrerak:** 000.00, martxa-sakagailua.

000.01, stop-sakagailua.

000.02, metal-detektagailua.

000.03, neurri-detektagailua.

000.04, A errele termikoa.

000.05, B errele termikoa.

**Irteerak:** 10.01, A motorra, A garraio-zinta.

10.00, B motorra, B garraio-zinta.

10.02, lixagailua.

**Tenporizadoreak:** TIM 000, segundo bat.

TIM 001, segundo bat.

### Programa:

LD 000.00

OR 10.01

AND NOT 000.01

AND NOT 000.04

OUT 10.01

LD NOT 000.02

LD TIM 001

AND 000.02

OR LD

AND 10.01

AND NOT 000.05

OUT 10.00

LD 000.02

AND 10.01

TIM 000


#0010

LD TIM 000

```

AND NOT 000.03
AND NOT 10.00
OUT 10.02
LD 000.03
TIM 001
    #0010
END(01)
    
```

**Eskema:**



**Funtzionamendua:**

Martxa-sakagailua (000.00) sakatzean, 10.00 (B motorra, B garraio-zinta) eta 10.01 (A motorra, A garraio-zinta) martxan jartzen dira. Metal-detektagailuak pieza bat detektatzen duenean, 10.00 (B motorra) itzali egiten da eta TIM 000 tenporizadoreak segundo bat kontatzen du. Orduan, lixagailua (10.02) martxan jartzen da. Piezak behar besteko neurriak dituztenean, neurri-detektagailua (000.03) aktibatu egiten da. Une horretan, lixagailua gelditu eta bere hasierako posiziora itzultzen da; aldi berean, tenporizadoreak (TIM 001) segundo bat kontatzen du. Denbora-tarte hori pasa ostean, B motorrak (10.00) martxan jarrarazten du berriro B garraio-zinta. Metal-detektagailuak beste pieza bat detektatzen duenean, prozesua errepikatu egiten da. A garraio-zinta martxan da denbora guztian; stop-sakatzailea edo errele termikoa aktibatuz gero baino ez da geldituko.

Stop-sakagailua sakatuz gero, bi motorrak gelditzen dira.

Motor bakoitzak bere errele termikoa du.

## 4 PRAKTIKA. KONTAGAILUAK, CNT INSTRUKZIOA

Lehenik eta behin, gogoratu behar da tenporizadoreek eta kontagailuek automataren memoria-eremu berbera erabiltzen dutela. Beraz, ezin dira programatu tenporizadoreak eta kontagailuak izen bera erabilia. Hau da, programa berean ezin dira programatu TIM 000 eta CNT 000.

CNT instrukzioak kontagailu baten moduan funtzionatzen du. Kontagailua programatzeko, bi datu behar dira: alde batetik, kontagailuaren izena (CNT 000, CNT 001, CNT 123,...), eta bestetik, zenbat pultsu kontatu behar dituen bere kontaktua itxi aurretik. Zenbaki hori programatzeko, “#” jarri behar da aurrean; adibidez, lau pultsu kontatzeko, #0004 idatzi behar da.

Kontagailuak bi sarrera ditu: batetik, pultsuak sartzen dira; bestearen bitartez, kontagailua kontatzen has daiteke berriro (RESET).

### 7. adibidea.

Lanpara bat kontrolatzea da helburua. Etengailua lau aldiz ixtean —hau da, lau pultsu sartzean—, piztu egin behar du lanparak.

**Sarrera:** 000.00, etengailua, pultsuak sartzekoa.

**Reset:** 000.01

**Irteera:** 10.00, lanpara.

**Kontagailua:** CNT 000.

**Programa:**

LD 000.00

LD 000.01

CNT 000


#0004

LD CNT 000

OUT 10.00

END(01)

Eskema:



Etengailua ixten den bakoitzean, kontagailuak pulsu bat gutxiago du memorian. Zerora heltzen denean, kontaktua itxi eta lanpara piztu egiten da. Pulsuak kontatzen amaitu baino lehen, RESET etengailuan pulsu bat sartuz gero, kontagailua berriro hasteko prest gelditzen da; hau da, beste lau pulsu sartu behar dira kontaktua ixteko.

## 4.1 Ariketak

### 16. ariketa

Zer egiten du zirkuitu honek?

**Sarrera:** 000.00, etengailua.

**Irteera:** 10.00, lanpara.


**Tenporizadorea:** TIM 000, bost segundo.

**Kontagailua:** CNT 001, hiru pultsu.

**Programa:**

```
LD 000.00
AND NOT TIM 000
TIM 000
    #0050
LD TIM 000
LD 000.01
CNT 001
    #0003
LD CNT 001
OUT 10.00
END(01)
```

**Eskema:**



0 etengailua itxi eta hamabost segundora, piztu egiten da lanpara. 0 etengailua ixtean, tenporizadorea kontaktzen hasten da. Bost segundo pasa ostean, TIM 000 kontaktua itxi eta kontagailuan pultsu bat sartzen da. Aldi berean, tenporizadorea berriro hasten da kontaktzen (RESET). Kontagailua hiru pultsu kontatzeko programatu denez gero, zikloa hirutan errepikatzen da. Hirugarren pultsuarekin batera, kontagailuaren kontaktua (CNT 001) itxi eta lanpara piztu egiten da.

## 17. ariketa

Lantegi batean metalezko piezak egiten dituzte. Langileek, pieza bat amaitu ahala, andel baten barruan uzten dute. Hortik aurrera, martxan jartzen da piezak kutxetan sartzeko prozesu automatikoa. Prozesua hau da:

- Andeleko atea irekitzean, astiro-astiro, malda txiki batean behera erortzen dira piezak.
- Bide horretan prest egon behar dute metal-detektagailua eta robotak, eta kutxak ondo kokatuta.
- Metal-detektagailuak pieza bat sumatzen duenean, robotak martxan jarri eta kutxan sartzen du pieza hori.
- Kutxa bakoitzean sei pieza sartzen dira. Kutxa betetzean —hau da, seigarren pieza sartzean—, beste robot batek beteta dagoen kutxa kendu eta hutsik dagoen beste bat jartzen du. Hori guztia egiteko, hiru segundo behar ditu robotak. Beteta dagoen kutxa, kamioira eramateko prest uzten du robotak, garraio-zinta baten gainean.
- Garraio-zintaren gainean zortzi kutxa beteta daudenean, martxan jartzen da zinta. Aldi berean, andeleko atea itxi egiten da, beste piezarik ez ateratzeko. Garraio-zintak hamar segundo ematen ditu martxan, eta, gero, gelditu egiten da.
- Prozesuari berriro ekiteko, ireki egin behar da andeleko atea.

**Sarrerak:** 000.00, martxa-sakagailua, andeleko atea irekitzen du.  
000.01, metal-detektagailua, piezak sumatzen ditu.

**Irteerak:** 10.00, andeleko atea irekita dagoela adierazten du.  
10.01, robotak pieza bat kutxara sartu duela adierazten du.  
10.02, kutxa-aldaketa adierazten du.  
10.03, garraio-zinta martxan dagoela adierazten du.

**Tenporizadoreak:** TIM 000, hiru segundo, kutxa-aldaketa.  
TIM 001, hamar segundo, garraio-zinta martxan.


**Kontagailuak:** CNT 002, sei pultsu, sei pieza kutxa bakoitzean.  
CNT 003, zortzi pultsu, zortzi kutxa beste garraio-zintan.




**Programa:**

```
LD 000.00
OR 10.00
AND NOT TIM 001
AND NOT CNT 003
OUT 10.00
LD 10.00
AND 000.01
AND NOT TIM 000
OUT 10.01
LD 10.01
LD CNT 002
CNT 002
    #0006
LD CNT 002
OR 10.02
AND NOT TIM 000
OUT 10.02
TIM 000
    #0030
LD CNT 002
LD CNT 003
CNT 003
    #0008
LD CNT 003
OR 10.03
AND NOT TIM 001
OUT 10.03
TIM 001
    #0100
END(01)
```

Eskema:





### Funtzionamendua:

Martxan jartzeko sakagailua (000.00) sakatzean, atea (10.00) ireki egiten da. Atea irekita dagoenean, piezak erortzen hasten dira. Metal-detektagailuak (000.01) pieza bat sumatzen duenean, robotak (10.01) kutxan sartu, eta kontagailuak (CNT 002) zenbatu egiten du. Kontagailu horrek (CNT 002) sei pieza kontatzen dituenean, beste robotak (10.02) kutxa aldatzen du hiru segundoan (TIM 000). Aldi berean, beste kontagailuak (CNT 003) kontuan hartzen du kutxa bat bete egin dela. Kontagailu horrek (CNT 003), zortzi kutxa bete kontatzen dituenean, atea itxi eta garraio-zinta martxan jartzen da hamar segundoan (TIM 001). Prozesuari berriro ekiteko, martxa-sakagailua (000.00) aktibatu behar da.

## 5 PRAKTIKA

### 5.1 Barne-erreleak

Lehenengo praktiketan esan dugun moduan, OMRON CPM2A automata programagarriak IR memoria-eremua erabiltzen du sarrerek eta irteerak programatzeko. IR memoria-eremuak, sarrerek eta irteerak programatzeaz gain, beste helbide erabilgarri batzuk ditu. IR 020tik IR 049ra bitarteko helbideak eta IR 200etik IR 227ra bitartekoak barne-errele moduan erabil daitezke. Barne-errele horiek ez dira fisikoak, baina oso tresna erabilgarriak dira zirkuitu baten programazioa egiteko.

### 8. adibidea.

Barne-errelea erabiliz, martxan jarri behar da tenporizadorea.

**Sarrerek:** 000.00, martxan jartzeko sakagailua.

000.01, stop-sakagailua.

**Irteera:** 10.00, lanpara.


**Tenporizadorea:** TIM 000, hiru segundo.

**Barne-errelea:** 20.00.

#### Programa:

```
LD 000.00
OR 20.00
AND NOT 000.01
OUT 20.00
LD 20.00
TIM 000
    #0030
LD TIM 000
OUT 10.00
END(01)
```

**Eskema:**



**Funtzionamendua:**

Martxa-sakagailua sakatzean, aktibatuko egiten da barne-errelea. Horrekin batera, tenporizadorea kontaktzen hasten da. Hiru segundo pasa ostean, TIM 000 kontaktua itxi eta lanpara piztu egiten da.

Automatek, sarrerek eta irteerak aktibatuta ote dauden adierazteko, LED argiak erabiltzen dituzte. Hau da, sarrera bat aktibatzen denean (etengailu, sentso edo sakagailu baten bidez), sarrera horri dagokion LED argia piztu egiten da. Hartara, zein sarrera dagoen aktibatuta ikus daiteke. Gauza bera gertatzen da irteeretan. Zirkuituak eraginda irteera bat aktibatzen denean (lanpara, motorra...), irteera horri dagokion LED argia piztu egiten da. Barne-erreleak, berriz, barneko funtzionamenduan erabiltzen ditu automatak, eta, beraz, ez dute seinaleztapenik (ez dira ikusten fisikoak ez direlako).

## 5.2 Barne-errele memoriadunak

OMRON CPM2A automata programagarriak HR memoria-eremua erabiltzen du barne-errele memoriadunak programatzeko. Hori egiteko helbide baliagarriak hauek dira: HR 00tik HR19ra bitartekoak.

Barne-errele horiek memoria dute. Hori da, hain zuzen ere, haien ezaugarri nagusia. Automataren funtzionamendua aldatu edo, argindarrak huts egiten badu ere, memorian zeukaten datua (ONen edo OFFen) gorde egiten dute erreleek.

### 8.1. adibidea

**Sarrera:** 000.00, martxa-sakagailua.


**Irteera:** 10.00, lanpara.

**Barne-errele memoriaduna:** HR 00.

**Programa:**

```
LD 000.00
OR HR 00
OUT HR 00
LD HR 00
OUT 10.00
END(01)
```

**Eskema:**



**Funtzionamendua:**

Martxa-sakagailua sakatzean, barne-errele memoriaduna aktibatu eta, horren ondorioz, lanpara piztu egiten da. Kontaktua irekiz gero, barne-errele memoriaduna aktibatuta (ON) egongo da, bere kontaktuaren bidez berrelikatzen da eta (HR 00). Posizio horretan automata deskonektatu eta konektatuz gero, lanpara piztu egiten da: izan ere, HR erreleak gorde egiten du aurretik zeukan informazioa (ON).

## 6 PRAKTIKA. OUT-OUT NOT, SET-RESET, KEEP(11), DIFU(13), DIFD(14)

Instrukzio eta funtzio horien bidez, helbide (lan-bit) baten zenbait kontrol egin daitezke.

### 6.1 OUT, OUT NOT

OUT instrukzioa irteerari dago lotuta, eta irteera hori sarreraren bitartez kontrolatzen da. Adibidez:

LD 000.00 (etengailua)

OUT 10.00 (lanpara)

END(01)

Etengailua irekita badago (OFF), lanpara itzalita egongo da (OFF).

Etengailua itxita badago (ON), lanpara piztuta egongo da (ON).

OUT NOT instrukzioak OUT instrukzioaren kontrako agindua ematen du, hau da:

LD 000.00 (etengailua)

OUT NOT 10.00 (lanpara)

END(01)

Etengailua irekita badago (OFF), lanpara piztuta egongo da (ON).

Etengailua itxita badago (ON), lanpara itzalita egongo da (OFF).

### 6.2 SET, RESET

Instrukzio horiek irteera bera kontrolatzeko erabiltzen dira. Instrukzio bakoitzak bere sarrera du, eta horren bitartez kontrolpean izaten da irteera. Adibidez:

LD 000.00, 0 kontaktua.

SET 10.00, lanpara.

LD 000.01, 1 kontaktua.

RESET 10.00, lanpara

END(01)

Lehen esan dugun moduan, SET instrukzioa eta RESET instrukzioa irteera berarekin (10.00, lanpara) daude erlazionatuta. Hala ere, bakoitza dagokion bitartez kontrolatu behar da: SET instrukzioa, 0 kontaktua bitartez; eta RESET instrukzioa, 1 kontaktua bitartez.

0 kontaktua ixtean (ON), lanpara piztu egiten da (10.00, ON). Posizio horretan 0 kontaktua irekiz gero (OFF), lanparak piztuta segitzen du. Hau da, lanpara piztuta dagoela, SET instrukzioari lotuta dagoen kontaktuak (0 kontaktua, 000.00) ez du eraginik. Lanpara itzaltzeko, RESET instrukzioari lotuta dagoen kontaktua (1 kontaktua, 000.01) itxi behar da. Lanpara berriro piztu ahal izateko, 1 kontaktua ireki behar da eta hasierako posiziora itzuli.

### 6.3 KEEP (11)

KEEP (11) funtzioak SET-RESET instrukzioek egiten duten eragiketa bera egiten du. KEEP (11) funtzioak irteera bat eta bi sarrera izan behar ditu erlazioatuta. Sarrera batek SET instrukzioarena egiten du; eta besteak, berriz, RESET instrukzioarena.

#### 9. adibidea.

Zer egiten du zirkuitu honek?

**Sarrerak:** 000.00, 0 kontaktua ( SET).  
000.01, 1 kontaktua (RESET).

**Irteera:** 10.00, lanpara.


**Barne-errelea:** 20.00

KEEP (11), 20.00

#### Programa:

```
LD 000.00
LD 000.01
KEEP (11) 20.00
LD 20.00
OUT 10.00
END(01)
```

#### Eskema:





**Funtzionamendua:**

0 kontaktua (SET) itxiz gero, KEEP (11) funtzioak bere irteera aktibatzen du, 20.00 barne-errelea, eta horrek pitzarazi egiten du lanpara. Posizio horretan, lanparak piztuta jarraitzen du, nahiz eta 0 kontaktua (SET) ireki. Lanpara itzaltzeko 1 kontaktua (RESET) itxi behar da. Horrek barne-errelea desaktibatu eta lanpara itzalarazten du. Lanpara piztu ahal izateko, 1 kontaktua ireki behar da eta hasierako posiziora itzuli.

## 6.4 DIFU(13), DIFD(14)

Funtzio horiek zer egiten duten aztertu baino lehen, scan-zikloa zer den azalduko dugu. Automatak prozesu bat kontrolatzen duenean, zuzen funtzionatzen duela ziurtatzeko, zenbait zeregin egin behar izaten ditu. Zeregin horiek ordenan daude ziklo baten barruan, scan-zikloaren barruan, hain zuzen. Scan-zikloa betetzeko automatak behar duen denbora-tartea zirkuituaren eta programaren arabera da; baina milisegundoak izan ohi dira.

DIFU(13) instrukzioak sarrera bat eta irteera bat ditu. Sarreran pulsu positiboa sartuz gero —hau da, kontaktua itxiz gero (OFFetik ONera)—, scan-zikloak irauten duen denbora berean egongo da aktibatuta irteera.

DIFD(14) instrukzioak eragiketa bera egiten du, baina irteera aktibatzeko pulsu negatiboa sartu behar da: hau da, sarrerako kontaktua ireki (ONetik OFFera).

## 18. ariketa

Ariketa honetan azkenengo praktiketari azalduko ditugun instrukzio eta funtzio erabiliko ditugu. (DIFU, HR, KEEP...)

**Sarrera:** 000.00, etengailua.

**Irteerak:** 10.00, motorra.

10.01, A irteera

10.02, B irteera

**Barne-erreleak:** 20.00

**Barne-errele memoriaduna:** HR 00

DIFU(13), 20.00

KEEP(11), HR 00

**Programa:**

LD 000.00

DIFU(13) 20.00

LD 20.00

AND NOT HR 00

OUT 10.01

LD 20.00

AND HR 00

OUT 10.02

LD 10.01

LD 10.02


KEEP(11), HR 00

LD HR 00

OUT 10.00

END (01)

Eskema:



**Funtzionamendua:**

- DIFU(13) funtzioa 20.00 barne-erreleari dago lotuta.
- Etengailua ixtean, 000.00 sarrera OFFetik ONera aldatzen da: hau da, pultsu positiboa sartzen da. Etengailua hasierako posizioan utzi behar da berriro.
- Pultsu positiboa DIFU(13) funtzioan sartzean, 20.00 barne-errelea aktibatu egiten da scan-zikloan.
- 20.00 barne-errelea scan-zikloan aktibatzen denean, 10.01 irteera piztu egiten da zikloak berak irauten duen bitartean.
- 10.01 irteera, aldi berean, KEEP(11) funtzioaren SET sarrera da. KEEP(11) funtzioa HR 00 barne-errele memoriadunarekin dago erlazionatuta. Hau da, 10.01 SET sarreratik pultsua sartzean, HR 00 aktibatu egiten da, eta horrek martxan jartzen du motorra (10.00).
- Motorra geldiarazteko, etengailua itxi egin behar da berriro. Hori egitean, pultsu positiboa sartzen da DIFU(13) funtzioan.
- 20.00 barne-errelea scan-zikloan aktibatuta dago. Horrek 10.02 irteera pizten du zikloak berak irauten duen bitartean.
- 10.02 irteera, aldi berean, KEEP(11) funtzioaren RESET sarrera da. KEEP(11) funtzioa HR 00 barne-errele memoriadunarekin dago erlazionatuta. Hau da, 10.02 RESET sarreratik pultsua sartzean, HR 00 desaktibatu eta motorra (10.00) gelditu egiten da.

Ariketa honetan DIFU(13) funtzioa erabili dugu. Funtzio hori pultsu positiboaz baliatzen da bere zeregina egiteko. DIFD(14) funtzioak berdin-berdin funtzionatzen du, baina pultsu negatiboa erabilia. Hau da, DIFD(14) funtzioak bere irteera aktibatzen du, sarrera ONetik OFFera pasatzen denean eta kontaktua irekitzean.

## 7 PRAKTIKA. INTERLOCK IL(02), INTERLOCK CLEAR ILC(03)

Instrukzio horiek beren artean dauden instrukzioetan eta funtzioetan dute eragina.

IL (02) instrukzioa sarrera baten bidez ONen jarriz gero, beraren eta ILC (03)ren artean programatuta dagoena bere horretan exekutatzen da.

IL (02) instrukzioa OFFen bada, beraren eta ILC (03)ren artean programatuta dagoena beste era batean exekutatzen da, instrukzio eta funtzioen arabera:

- OUT eta OUT NOT instrukzioen helbideak OFFen izango dira.
- TIM tenporizadoreak hasierako posiziora itzuliko dira (RESET).
- CNT kontagailuek eutsi egingo diote une horretan kontaktzen ari denari.
- DIFU(13) eta DIFD(14) instrukzioak ez dira exekutatuko.
- Beste instrukzioak ez dira exekutatuko, edo haien helbideak OFFen izango dira.

IL (02) funtzioak ILC (03) funtzioa hartzen du erreferentziatzat. Baina, hala ere, ez dira beti elkarrekin programatu behar. Programa baten barruan hainbat IL(02) jar daitezke ILC(03) bakarrarekin. Guztiek erreferentzia bera izango dute.

Era berean, ILC(03) bat baino gehiago programatzeko aukera ere badugu, baina inoiz ere ez bi jarraian. Haien artean IL(02) funtzioen bat txertatu behar da beti.

### 10. adibidea

Zer egiten du zirkuitu honek?

**Sarrerak:** 000.00, 0 etengailua, IL(02) instrukzioarena.

000.01, 1 etengailua, tenporizadorearena.

000.02, 2 etengailua, kontaketa egitekoa.

000.03, 3 etengailua, RESET.

000.04, 4 etengailua, lanpararena.

**Irteera:** 10.00, lanpara.

IL(02)-ILC(03)


TIM 000, bost segundo.

CNT 001, zortzi pultsu.

**Programa:**

```
LD 000.00
IL(02)
LD 000.01
TIM 000
    #0050
LD 000.02
LD 000.03
CNT 001
    #0008
LD 000.04
OUT 10.00
END(01)
```

**Eskema:**



**Funtzionamendua:**

Programa osoa IL(02) instrukzioaren mende dago: 000.00 sarrera itxiz gero, IL(02) ONen izango da. Orduan, programa bere horretan exekutatzen da:

- 000.01 sarrera ixten bada, tenporizadorea kontaktzen hasten da.
- 000.02 sarrera ixten den bakoitzean, kontagailuak pultsu bat zenbatzen du.
- 000.03 sarrera ixten denean, kontagailua hasierako posiziora itzultzen da.
- 000.04 sarrera ixten bada, lanpara piztu egiten da.

000.00 sarrerako etengailua irekitzen bada, IL(02) funtzioa OFFen izango da. Orduan,

- Tenporizadorea hasierako posiziora itzuli eta hor gelditzen da. Horrela, 000.01 sarrerak ez du eraginik.
- Kontagailuak eutsi egiten dio une horretan kontaktzen ari denari. Sarrerek ez dute eraginik.
- Lanpara itzali egiten da, eta, nahiz eta 000.04 sarrera itxi, ez da pizten.

## 8 PRAKTIKA. JMP(04)-JME(05)

Instrukzio horiek elkarrekin erabili behar dira beti. Hartara, programa baten barruan JMP(04)tik JME(05)era jauzi egiteko aukera dago.

JMP(04) ONen denean —hau da, sarrera itxita dagoenean—, programa bere horretan exekututzen da.

JMP(04) OFFen denean —hots, sarrera irekita dagoenean—, programak JMP(04)tik JME(05)era egiten du jauzi, eta tartean dauden instrukzioak ez dira exekututzen: dauden-daudenetan gelditzen dira.

Programa baten barruan jauzi bat baino gehiago egin daiteke. Zehazteko programak nondik nora egin behar duen jauzi, instrukzioak (JMP eta JME) izendatu behar dira. Jauzi-bikoteek zenbaki bera dute. Hau da, JMP(04) 02 eta JME(05) 02 elkarri lotuta egongo dira; edo JMP(04) 00 eta JME(05) 00; edo JMP(04) 05 eta JME(05) 05...


### 11. adibidea

**Programa:**

```
LD 000.00
JMP(04) 00
LD 000.01
TIM 000
    #0050
LD 000.02
LD 000.03
CNT 001
    #0005
LD 000.04
OUT 10.00
JME(05) 00
END(01)
```



Eskema:



Funtzionamendua:

- JMP(04) 00 ONen denean, programa bere horretan exekutatzen da.
- JMP(04) 00 OFFen denean, beraren eta JME(05) 00ren artean dauden instrukzioak ez dira exekutatzen.
- Zirkuitua martxan badabil eta JMP(04) 00 ONetik OFFera pasatzen bada, beraren eta JME(05) 00ren artean dauden instrukzioak bere horretan gelditzen dira (bai tenporizadorea, bai kontagailua, bai lanpara).

## 9 PRAKTIKA. DIAGNOSTIC FUNTZIOA. FAL(06), FALS(07)

Praktika honetan, DIAGNOSTIC funtzioak —FAL(06) eta FALS(07) funtzioak, alegia— azaldu eta erabiliko ditugu.

Zirkuituetan gerta daitezkeen hainbat akatsi aurre egiteko erabil daitezke funtzio horiek.

FAL(06) funtzioa akats arinei aurre egiteko erabiltzen da. FAL(06) funtzioa ONen jartzen bada, automataren ALARM/ERROR argia piztu-itzalika aritzen da, baina zirkuituak bere horretan jarraitzen du. Hau da, akatsen bat dagoelako abisua ematen du funtzio honek, baina ez du prozesua geldiarazten.

FALS(07) funtzioa akats larriei aurre egiteko erabiltzen da. FALS(07) funtzioa ONen jartzen bada, automataren ALARM/ERROR argia piztu eta prozesua geldiarazten da. Hau da, funtzio horrek prozesua geldiarazi eta abisua ematen du.

FAL(06) eta FALS(07) funtzioak nola erabili ondo ulertzeko, egokia da adibide hau:

Andel bateko likido-maila kontrolatzeko, lau sentzore ditugu. Gehienetan, bi maila zehatzen artean egongo da likidoa: goiko mailaren eta beheko mailaren artean, alegia.

Likidoak goiko maila gainditzen badu ere, artean ez dago andela guztiz betetzeko arriskurik, baina andela betetzen duen iturriari ixteko agindua ematen dio sentzoreak.

Likidoak beheko mailatik behera egiten badu ere, artean ez dago andela guztiz husteko arriskurik, baina andela betetzen duen iturriari irekitzeko agindua ematen dio sentzoreak.

Kasu horietan bietan, FAL(06) erabil daiteke: prozesua ez du geldiaraziko, baina abisatu egingo du.

Andela guztiz betetzeko arriskua dagoenean, beste sentzore bat aktibatzen da.

Beste sentzore bat jar daiteke, andela guztiz husteko arriskua dagoela abisatzeko.

Kasu horietan bietan, FALS(07) erabil daiteke: akats larria gertatu da eta, abisua emateaz gain, prozesua bera geldiarazi egingo du.

DIAGNOSTIC funtzioak zer diren azaldu ondoren, ariketa hau egitea planteatzen da:

Lantegi batean piezak kutxetan sartzeko, prozesu automatikoa erabiltzen dute: alde batetik, garraio-zintak kutxak eramaten ditu; bestetik, robotak piezak sumatu eta kutxan sartzeko dituzte. Prozesua hau da:

Sentsoreak kutxa sumatzen duenean, garraio-zinta gelditu egiten da. Bi segundo geroago, robota martxan jartzen da eta sumatzen dituen piezak kutxan sartzeko dituzte. Hamabi pieza kutxan sartu ondoren, robota gelditu egiten da. Bi segundo geroago, garraio-zinta martxan jartzen da berriro, harik eta sentsoreak beste kutxa bat sumatu arte.

Prozesu automatikoa diseinatu eta martxan jarri ere egin zuten; baina, batzuetan, arazoak izaten zituzten: noizean behin, kutzak edo piezak trabatuta gelditzen ziren, eta denbora alferrik galtzen zuten ohiko martxara itzuli arte. Arazo hori lehenbailehen konpontzeko, DIAGNOSTIC funtzioak erabiltzea erabaki zuten.

Sentsoreak hogeita hamar segundoan ez badu kutzarik sumatzen, FALS(07) instrukzioa ONen jartzen da: kutzak trabatuta daudela adierazi eta prozesua geldiarazten du kutzak ondo kokatu arte.


Era berean, robotak hamabost segundoan ez badu piezarik sumatzen, FALS(07) instrukzioa ONen jartzen da: piezak trabatuta daudela adierazi eta prozesua geldiarazten du piezak ondo kokatu arte.


- Sarrerak:** 000.00, martxa-sakagailua.  
000.01, sentsorea, kutzak detektatzeko.  
000.02, robotaren sentsorea, piezak detektatzeko.  
000.03, stop-sakagailua.  
000.04, errele termikoa.
- Irteerak:** 10.00, garraio-zinta.  
10.01, robota.
- Tenporizadoreak:** TIM 000, bi segundo.  
TIM 001, bi segundo.  
TIM 002, hogeita hamar segundo.  
TIM 003, hamabost segundo.
- Kontagailuak:** CNT 004, hamabi pultsu.
- Barne-errelea:** 20.00  
IL-ILC, FALS.

**Programa:**

```
LD 000.00
OR 20.00
AND NOT 000.03
AND NOT 000.04
OUT 20.00
LD 20.00
IL(02)
LD NOT 000.01
OR TIM 001
OUT 10.00
TIM 002
    #0300
LD 000.01
TIM 000
    #0020
LD TIM 000
AND 000.02
AND NOT CNT 004
OUT 10.01
LD 10.01
LD 000.01
AND 10.00
CNT 004
    #0012
LD CNT 004
AND NOT 000.01
TIM 001
    #0020
LD NOT 000.02
TIM 003
    #0150
LD TIM 002
FALS(07) 01
LD TIM 003
FALS(07) 02
ILC(03)
END(01)
```

Eskema:





**Funtzionamendua:**

Martxa-sakagailuari eragitean stop-sakagailua edo errele termikoa aktibatuta ez badaude, 20.00 barne-erreleak IL(02) instrukzioa ONen jartzen du, eta horrela prozesua hasteko baimena ematen du. Une horretan, kutxak eramateko garraio-zinta eta hogeita hamar segundo kontaktzen dituen tenporizadorea martxan jartzen dira. Hogeita hamar segundo horietan sentsorearen aurretik ez bada kutxarik pasatzen, kutxak trabatuta dauden seinale; orduan, FALS(07) ONen jartzen da eta, horren ondorioz, prozesua geldiarazi eta abisua ematen du.

Kutxak ez badaude trabatuta, prozesuak bere horretan segitzen du eta kutxak sentsorearen aurretik pasatzen dira. Sentsoreak kutxa sumatzen duenean, garraio-zinta geldiarazi, eta martxan jartzen da bi segundo kontaktzen dituen tenporizadorea. Bi segundo horiek pasa ostean, eta ez lehenago, robota lanean hasteko prest dago. Robotak, pieza sumatzen duenean, kutxan sartzen du, 10.01 irteera aktibatuta. Hau da, robotak pieza bat kutxan sartzen duen bakoitzean, 10.01 irteera aktibatu egiten da. Beraz, kontagailuak pultsuak zenbatzeko duen sarreran 10.01 irteera jarri behar da, horrela piezak kontatzeko aukera izateko. Hamabi pieza kutxan sartzen dituztenean, robota gelditu egiten da eta ezin du beste piezarik sumatu. Bi segundo geroago, garraio-zinta martxan jartzen da berriro.

Robotak pieza bat sumatzen duen bakoitzean, martxan jartzen da hamabost segundo kontaktzen dituen tenporizadorea. Hamabost segundo horietan robotak ez badu beste piezarik sumatzen, piezak trabatuta dauden seinale; orduan, FALS(07) ONen jartzen da eta, horren ondorioz, prozesua geldiarazi eta abisua ematen du.

## 10 ARIKETAK

### 19. ariketa

Lixagailu batek pieza neurtuak egiten ditu. Prozesua hau da:

- Lehenik eta behin, lixatzeko tokian jarri behar da pieza. Behar bezala jarrita dagoela ziurtatzeko, sentsore bi erabiliko ditugu. Sentsore bietako batek pieza gaizki kokatuta dagoela sumatzen badu, prozesuak ez du aurrera egingo.
- Lixatu beharreko pieza ondo kokatuta dagoela adierazten dutenean sentsoreek, martxa-sakagailua saka daiteke. Une horretan, martxan jartzen da prozesua. Pieza behar bezala jarri gabe dagoenean martxa-sakagailua sakatzen badugu, prozesua ez da martxan jartzen. Era berean, behin prozesua martxan dagoela, sentsoreek pieza mugitu dela sumatzen badute, prozesua eten egiten da.
- Martxa-sakagailua sakatzean, bi motor jartzen dira martxan. Bata, lixagailuarena berarena da; besteak lixagailua mugiarazten du: piezara hurbiltzen eta piezatik aldentzen du. Hasieran, lixagailua martxan jarrarazi eta piezara hurbilarazten dute motorrek.
- Lixagailuak bere zeregina betetzen du. Sentsore batek egiaztatzen du piezak behar besteko neurriak dituela. Une horretan, lixagailuaren motorra geldiarazi eta beste motorrak piezatik aldenarazten du.
- Beste sentsore batek lixagailua hasierako lekura heldu dela sumatu eta motorra geldiarazten du. Horrela amaitzen da prozesua.
- Zikloari berriro ekiteko, pieza aldatu eta martxa-sakagailua sakatu behar da.
- Larrialdietako sakagailuak eten egiten du prozesua.

**Sarrerak:** 000.00, 0 sentsorea, pieza ondo kokatuta dagoela ziurtatzeko.  
 000.01, 1 sentsorea, pieza ondo kokatuta dagoela ziurtatzeko.  
 000.02, larrialdietako sakagailua.  
 000.03, martxa-sakagailua.  
 000.04, 4 sentsorea, piezak behar besteko neurriak dituela ziurtatzeko.  
 000.05, 5 sentsorea, zikloa amaitzeko.

**Irteerak:** 10.00, 0 motorra, lixagailua piezara hurbilarazteko.  
 10.01, 1 motorra, lixagailuarena.  
 10.02, 0 motorra, lixagailua piezatik aldenarazteko.


**Barne-erreleak:** 20.00  
 20.01  
 20.02  
 20.03



**Programa:**

LD 000.00  
AND 000.01  
AND NOT 000.02  
IL(02)  
LD 000.03  
OR 20.00  
AND NOT 20.03  
OUT 20.00  
LD 20.00  
IL(02)  
LD 000.04  
OR 20.01  
OUT 20.01  
LD 000.05  
OR NOT 20.01  
AND NOT 20.02  
OUT 10.01  
AND NOT 10.02  
OUT 10.00  
LD 20.01  
AND NOT 000.05  
AND NOT 10.00  
OUT 10.02  
LD 000.05  
AND NOT 000.04  
OR 20.02  
OUT 20.02  
LD 000.05  
AND 20.02  
OR 20.03  
OUT 20.03  
ILC(03)  
END(01)

Eskema:



**Funtzionamendua:**

Pieza behar bezala jarrita dagoela adierazten duten 0 sentsorea eta 1 sentsorea aktibatzen direnean, eta larrialdietako sakagailua aktibatuta ez badago, IL(02) ONen izango da. Horrek aukera ematen du martxa-sakagailua erabiltzeko. Sentsoreren bat aktibatuta ez dagoelako edo larrialdietarako sakagailua sakatu dugulako IL(02) OFFen bada, martxan jartzeko sakagailuak ez du eraginik izango.

IL(02) ONen bada, martxa-sakagailua sakatzean, barne-errele bat aktibatzen da. Horrek ONen jartzen du beste IL(02) bat, eta IL(02) horrek baimena ematen du prozesua hasteko. Baimen hori emateko, lixagailuak hasierako posizioan egon behar du. Hau da, prozesua hasi ahal izateko, baldintza batzuk bete behar ditu zirkuituak. Lehenengo, IL(02)ren bidez, kontrolatu behar da pieza ondo jarrita dagoen eta larrialdietako sakagailua aktibatu gabe dagoen; eta bigarren, IL(02)ren bidez, berriz, lixagailua hasierako posizioan dagoen.

Hori dena ondo badago, martxan jartzen dira motorrak. Piezak behar bezain neurri zehatzak ditutenean, lixagailuaren motorra gelditu eta beste motorrak alderantzizko noranzkoan biratzen du, hasierako posiziora itzultzeko. Horra heltzean, motorra itzali eta zikloa bukatu egiten da.

Etaparen guztien kontrola egiteko, barne-erreleak erabili behar dira.

## 20. ariketa

Zer egiten du zirkuitu honek?

**Sarrerak:** 000.00, martxa-sakagailua.

000.01, stop-sakagailua.

000.02, etengailua.

**Irteerak:** 10.00, A motorra.

10.01, B motorra.

**Barne-errelea:** 20.00

**Tenporizadoreak:** TIM 000, bost segundo.

TIM 001, bost segundo.

IL-ILC, JMP-JME.

### **Programa:**

LD 000.00

OR 20.00

AND NOT 000.01

OUT 20.00

LD 20.00

IL(02)

LD 000.02

JMP(04) 00

LD NOT TIM 001

TIM 000

#0050

LD TIM 000

TIM 001

#0050

JME(05)

LD TIM 000


OUT 10.00

OUT NOT 10.01

ILC(03)

END(01)

Eskema:



### Funtzionamendua:

Martxa-sakagailua sakatzean, barne-erreleak aktibatu eta IL(02) ONen jartzen du. Horrek baimena ematen du gainontzeko instrukzioak exekutatzeko.

Bestalde, JMP(04) funtzioa dago. Funtzio horri dagokion etengailua itxita dagoen bitartean, ez du jauzirik egiten eta programa bere horretan exekutatzen da: hau da, motor bat martxan jarri eta bestea geldiarazi egiten du. Bost segundo pasa ostean, martxan zegoena geldiarazi eta bestea martxan jarrarazten du. Beste bost segundo pasa ostean, berriro aldarazten ditu motorrak. Eta horrela segitzen du lanean zirkuituak.

Stop-sakagailua sakatuz gero, bi motorrak gelditu egiten dira.

JMP(04) funtzioaren etengailua (000.02 sarrera) irekiz gero, programak JMP(04)tik JME(05)era egiten du jauzi, eta tartean dauden tenporizadoreek eutsi egiten diete une horretan kontatzen ari direnei. Hots, prozesua gelditu egiten da, baina posizioak bere horretan segitzen du.

## 21. ariketa

Lantegi batean egurrezko altzariak egiten dituzte. Langileek zerra mekanikoa erabiltzen dute lanean. Zerra mekanikoak hiru botoi ditu: stop-sakagailua, eskuinaldera zerratzeko botoia eta ezkerraldera zerratzeko botoia. Zerrak horrela funtzionatzen du:

Botoiaren bidez eskuinaldera zerratzeko agindua emanaz gero, zerrak alde horretara egiten du. Muturrera heltzen denean, hamar segundoz gelditu egiten da. Denbora-tarte hori pasatakoan, ezkerraldera egiten du zerrak. Ezkerreko muturrera heltzen denean, hamar segundoz gelditu eta eskuinaldera abiatzen da berriro.


Muturrera heldu baino lehen zerraren noranzkoa aldatu nahi izanez gero, lehenik eta behin geldiarazi egin behar da zerra stop-sakagailuari eraginda. Eskuinaldera eta ezkerraldera zerratzeko botoiek baino ez dute eragina zerra geldirik dagoenean.

- Sarrera:** 000.00, eskuinaldera zerratzeko botoia.  
000.01, ezkerraldera zerratzeko botoia.  
000.02, eskuineko muturreko ibiltarteko etengailua.  
000.03, ezkerreko muturreko ibiltarteko etengailua.  
000.04, stop-sakagailua.  
000.05, errele termikoa.
- Irteerak:** 10.00, eskuinaldera abiatzeko.  
10.01, ezkerraldera abiatzeko.
- Tenporizadoreak:** TIM 000, hamar segundo  
TIM 001, hamar segundo.

**Programa:**

```
LD NOT 000.04
AND NOT 000.05
IL(02)
LD 000.00
AND NOT 000.01
OR 10.00
OR TIM 001
AND NOT 000.02
AND NOT 10.01
OUT 10.00
LD 000.01
AND NOT 000.00
OR 10.01
OR TIM 000
AND NOT 000.03
AND NOT 10.00
OUT 10.01
LD 000.02
TIM 000
    #0100
LD 000.03
TIM 001
    #0100
ILC(03)
END(01)
```

Eskema:





**Funtzionamendua:**

Stop-sakagailua edo errele termikoa aktibatuta badaude, zirkuituak ez du funtzionatzen. Horretarako, IL(02) funtzioa erabiltzen da. Stop-sakagailua eta errele termikoa aktibatu gabe daudenean, IL funtzioa ONen da, eta zirkuitua funtzionatzeko prest dago. Posizio horretan, aukera dago eskuinaldera nahiz ezkerraldera zerratzeko. Nahi den aldea aukeratu eta zerrak horra jotzen du. Muturrera heltzean, motorra gelditu eta tenporizadorea martxan jartzen da. Hamar segundo pasa ostean, tenporizadorearen kontaktuaren eraginez beste aldera egiten du zerrak.

Zerra alde batera mugitzen ari bada, beste aldera zerratzeko agindua emateko botoiak ez du eragirik. Lehenik eta behin, geldiarazi egin behar da zerra stop-sakagailua sakatuta, eta gero eskuinaldera edo ezkerraldera aukeratu.

## 22. ariketa

Ariketa honen helburua haizagailu bat kontrolatzea da. Haizagailuak eskuinaldera edo ezkerraldera egin dezake bira: Haizagailua alde batera biratzen ari bada, beste aldera birarazi ahal izateko, ezinbestekoa da geldi-geldi egotea. Stop-sakagailuari eragitean, haizagailuaren motorra gelditu egiten da, baina une horretan zuen abiaduraren ondorioz, haizagailuak hamabost segundo behar ditu guztiz gelditzeko. Denbora-tarte hori pasatu artean ez du eraginik beste aldera biratzeko botoiak.

Hau da, geldi-geldi dagoenean bakarrik agindu diezaiokegu haizagailuari ezkerraldera edo eskuinaldera biratzeko. Haizagailua martxan badago eta beste aldera biratzea nahi badugu, stop-sakagailuari eragin behar zaio lehenik eta behin. Hamabost segundo pasatu ondoren, haizagailua erabat gelditu denean, beste aldera biratzeko botoia sakatu eta motorra martxan jartzen da.

Zirkuitua gainkargetatik babesteko, errele termikoa erabili behar da. Errele termikoa aktibatzen bada, seinaleztapena ere martxan jartzen da. Ikusteko eta entzuteko modukoa da seinaleztapena: ikustekoa finkoa da, eta entzutekoak, berriz, bi segundo ematen ditu joka.

**Sarrerak:** 000.00, eskuinaldera biratzeko botoia.

000.01, ezkerraldera biratzeko botoia.

000.02, stop-sakagailua.

000.03, errele termikoa.

**Irteerak:** 10.00, eskuinalderako bira.

10.01, ezkerralderako bira.

10.02, ikusteko seinaleztapena.

10.03, entzuteko seinaleztapena.

**Tenporizadoreak:** TIM 000, hamabost segundo.

TIM 001, hamabost segundo

TIM 002, bi segundo.


**Barne-erreleak:** 20.00, 20.01, 20.02, 20.03


DIFD(14)

**Programa:**

```
LD 000.00
AND NOT 000.01
OR 10.00
AND NOT 000.02
AND NOT 000.03
AND NOT 10.01
AND NOT 20.03
OUT 10.00
DIFD(14) 20.00
LD 000.01
AND NOT 000.00
OR 10.01
AND NOT 000.02
AND NOT 000.03
AND NOT 10.00
AND NOT 20.02
OUT 10.01
DIFD(14) 20.01
LD 20.00
OR 20.02
AND NOT TIM 000
OUT 20.02
TIM 000
    #0150
LD 20.01
OR 20.03
AND NOT TIM 001
OUT 20.03
TIM 001
    #0150
LD 000.03
OUT 10.02
TIM 002
    #0020
AND NOT TIM 002
OUT 10.03
END(01)
```

Eskema:





**Funtzionamendua:**

Haizagailua geldi-geldirik badago, eskuinalderako bira edo ezkeralderakoa aukeratu, eta motorrak halaxe egiten du.

Martxan dabilen bitartean bat-batean beste aldera bira ez dezan, babestu egin behar da haizagailua. Horretarako, kontaktuak berak erabili behar dira eskuinaldera eta ezkeraldera biratzeko.

Stop-sakagailuari eragitean, motorra geldiarazteaz gain, DIFD(14) instrukzioak pulstu negatiboa hartzen du, eta, horrela, dagokion barne-errelea aktibatu. Barne-errele horren bitartez, tenporizadorea martxan jartzen da eta hamabost segundoan ez dio beste aldera biratzeko aukerarik ematen haizagailuari; hartara, guztiz gelditzeko beta ematen dio.

Errele termikoa aktibatzen bada, ikusteko seinaleztapena piztu egiten da. Aldi berean, tenporizadore baten bidez kontrolatzen da entzuteko seinaleztapena.

## 23. ariketa

Zama-jasogailu bat kontrolatzea da ariketa honen helburua. Betebeharrak nolakoak diren, automatikoki edo langile batek aginduta funtzionatzen du zama-jasogailuak.

Kommutadore baten bidez kontrolatzen da zama-jasogailua. Kommutadoreak hiru posizio ditu: automatikoa, eskuzkoa eta stop.

Kommutadorea eskuzko posizioan jartzean, langileak agindu behar dio zama-jasogailuari igotzeko, jaisteko edo gelditzeko. Martxan dagoenean zama-jasogailuaren mugimendua aldarazten saiatuz gero (igo-jaitsi / jaitsi-igo), zirkuituak ez du kasurik egiten. Lehenik eta behin, gelditzeko botoia sakatu behar da eta mugimendua eten; eta horren ondoren, igotzeko edo jaisteko agindua eman.

Kommutadorea posizio automatikoan jarritz gero, zama-jasogailua behean jarri behar da, eta ziklo hau egiten du:

- Hamabost segundoan igo egiten da, duen karga gora eramateko.
- Hogei segundoan geldi gelditzen da, langileak huts dezan.
- Hamar segundoan jaitsi egiten da, hutsik.
- Hogeita hamar segundoan geldi gelditzen da, langileak karga dezan.

Zikloa errepikatu egiten da.

**Sarrerak:** 000.00, posizio automatikoa.

000.01, eskuzko posizioa.

000.02, igotzeko sakagailua

000.03, jaisteko sakagailua.

000.04, gelditzeko.

000.05, errele termikoa.

**Irteerak:** 10.00, zama-jasogailua gora.

10.01, zama-jasogailua behera.

**Tenporizadoreak:** TIM 000, hamabost segundo

TIM 001, hogeita hamabost segundo

TIM 002, berrogeita bost segundo

TIM 003, hirurogeita hamabost segundo


**Barne-erreleak:** 20.00, 20.01

DIFU(13), IL(02)-ILC(03)


**Programa:**

LD 000.00  
OR 000.01  
DIFU(13) 20.00  
LD 20.00  
OR 20.01  
AND NOT 000.05  
OUT 20.01  
LD 20.01  
IL(02)  
LD 000.00  
AND NOT TIM 000  
LD 000.02  
OR 10.00  
AND NOT 000.04  
AND 000.01  
OR LD  
AND NOT 10.01  
OUT 10.00  
LD 000.00  
AND TIM 001  
AND NOT TIM 002  
LD 10.01  
OR 10.01  
AND NOT 000.04  
AND 000.01  
OR LD  
AND NOT 10.00  
OUT 10.01  
LD 000.00  
AND NOT TIM 003  
TIM 000  
    #0150  
TIM 001  
    #0350  
TIM 002  
    #0450  
TIM 003  
    #0750  
ILC(03)  
END(01)

Eskema:







### Funtzionamendua:

Zirkuituak funtzionatu ahal izateko, eskuzko posizioa edo automatikoa aukeratzean, DIFU(13) funtzioak pulsu positiboa jasotzen du, eta horrela 20.00 barne-errelea aktibatuz. Errele termikoa ez badago aktibatuta, 20.01 barne-erreleak IL(02) funtzioa ONen jarri, eta zirkuituari programatu bezala funtzionatzeko baimena ematen dio.

Eskuzko posizioa aukeratuz gero, lehen azaldu dugun moduan funtzionatzen du zirkuituak. Langileak kontrolatzen du zama-jasogailua, eta igotzeko edo jaisteko agindua eman diezaiolke zama-jasogailuari, beti ere geldidagoenean.

Posizio automatikoa aukeratuz gero, zama-jasogailua igotzen hasten da. Aldi berean, martxan jartzen dira lau tenporizadoreak. Lau tenporizadoreon bidez kontrolatzen zama-jasogailuaren zikloa (gorago aipatu da hori).

## 24. ariketa

Garaje bateko ate automatikoen horrela funtzionatu behar dute:

Irekitzeko agindua ematean, atea gora egin behar du. Goiko posiziora heltzen denean, kontaktuak itxi eta motorrak gelditu egin behar du. Ateak hogeituro segundo eman behar ditu irekita eta gero, itxen hasi. Beheko posiziora heltzen denean, kontaktuak itxi eta motorrak gelditu egin behar du.

Atea itxen den bitartean, sentsore batek oztoporik ez dagoela ziurtatu behar du. Sentsoreak ate azpian zerbait dagoela sumatzen badu, berehala geldiarazi behar du motorra. Ateak, segundo batean geldirik egon ostean, irekitzen hasi behar du, goiko kontaktuak motorra geldiarazi arte.

Atea irekitzeko eta ixteko agindua, garajearen barrutik edo kanpotik eman daiteke. Garajearen barruan gaudenean, ateko botoia erabili behar da; kanpoan gaudenean, ostera, urrutiko agintea.

Atea irekitzeko agindua ematean, garajeko argia piztu egingo da, eta bi minutu emango ditu piztuta.

**Sarrerak:** 000.00, urrutiko agintea.

000.01, barruko botoia.

000.02, goiko kontaktua, atea guztiz irekita dagoenean motorra geldiarazten duena.

000.03, beheko kontaktua, atea guztiz itxita dagoenean motorra geldiarazten duena.

000.04, sentsorea.

**Irteerak:** 10.00, atea irekitzeko.

10.01, atea ixteko.

10.02, garajeko argia.

**Tenporizadoreak:** TIM 000, hogeituro segundo.


TIM 001, segundo bat.


TIM 002, ehun eta hogeituro segundo.

**Programa:**

```
LD 000.00
OR 000.01
OR TIM 001
OR 10.00
AND NOT 10.01
AND NOT 000.02
OUT 10.00
LD 000.00
OR 000.01
OR TIM 000
OR 10.01
AND NOT 10.00
AND NOT 000.03
AND NOT 20.00
OUT 10.01
LD 000.02
TIM 000
    #0200
LD 10.01
OR 20.00
AND 000.04
OUT 20.00
TIM 001
    #0010
LD 000.00
OR 000.01
OR 10.02
AND NOT TIM 002
OUT 10.02
TIM 002
    #1200
END(01)
```

Eskema:





**Funtzionamendua:**

Atea itxita dagoenean, irekitzeko agindua ematen badugu, motorrak atea altxatzen du. Atea gora heltzen denean, kontaktuak motorra geldiarazten du eta TIM 000 tenporizadorea martxan jarrarazten du. Hogei segundo pasa ostean, tenporizadorearen kontaktuak aktibatua eta atea itxi egiten da.

Atea ixten ari den bitartean, sentsoreak oztoporen bat sumatzen badu, geldiarazi egiten du motorra 20.00 barne-errelearen bidez. Aldi berean, TIM 001 tenporizadorea martxan jartzen da, kontaktuak segundo bat geroago itxi eta atea irekitzen hasten da, goiko posiziora heldu arte.

Atea irekitzeko agindua ematean, piztu egiten da garajeko argia. Aldi berean, TIM 002 tenporizadorea martxan jartzen da, kontaktuak 2 minutu geroago itxi eta argia itzali egiten da.

## 25. ariketa

Semaforo bat kontrolatzea da helburua. Semaforoak horrela funtzionatu behar du:

Semaforoaren zikloa martxan jartzeko botoia irekita badago, anbar-koloreko argia piztu-itzalika ibiliko da gidarientzat. Oinezkoentzako semaforoa, berriz, itzalita egongo da.

Semaforoa martxan jartzeko botoia itxiz gero, zikloa hau izango da:

- Bi minutuan: gidarientzat argi berdea, eta oinezkoentzat, aldiz, gorria.
- Hamar segundoan: gidarientzat anbar-koloreko argia, eta oinezkoentzat, aldiz, gorria.
- Hogeita hamar segundoan; gidarientzat argi gorria, eta oinezkoentzat, aldiz, berdea.
- Hamar segundoan; gidarientzat gorria, eta oinezkoentzat, aldiz, piztu-itzalika argi berdea.
- Zikloaren hasiera.

**Sarrera:** 000.00, martxa.

**Irteerak:** 10.00, gidarientzako argi berdea.

10.01, gidarientzako argi gorria.

10.02, gidarientzako anbar-koloreko argia.

10.03, oinezkoentzako argi berdea.

10.04, oinezkoentzako argi gorria.

**Tenporizadoreak:** TIM 000, ehun eta hogeit hamar segundo.

TIM 001, ehun eta hogeita hamar segundo.

TIM 002, ehun eta hirurogei segundo.

TIM 003, ehun eta hirurogeita hamar segundo.


TIM 004, segundo erdia.

TIM 005, segundo erdia.


**Programa:**

```
LD 000.00
AND NOT TIM 003
TIM 000
    #1200
TIM 001
    #1300
TIM 002
    #1600
TIM 003
    #1700
LD 000.00
AND NOT TIM 001
OUT 10.04
AND NOT TIM 000
OUT 10.00
LD TIM 000
AND NOT TIM 001
LD NOT 000.00
AND NOT TIM 004
OR LD
OUT 10.02
LD TIM 001
OUT 10.01
AND NOT TIM 004
OUT 10.03
LD TIM 002
OR NOT 000.00
AND NOT TIM 005
TIM 004
    #0005
AND TIM 004
TIM 005
    #0005
END(01)
```

Eskema:









LANBIDE  
EKIMENA

